```
SetOptions[EvaluationNotebook[], CellContext → Notebook]
```

Curves in 3D

Exmaple: the Helix

```
r[t_] := {2 Cos[t], 3 Sin[t], t};
```

```
Simplify[FrenetSerretSystem[r[t], t], Assumptions → Element[t, Reals]]
```

$$\left\{\left\{\frac{2\sqrt{17-\text{Cos}[2\,t]}}{5\,(3+\text{Cos}[2\,t])^{3/2}},\ -\frac{12}{5\,(-17+\text{Cos}[2\,t])}\right\},\right.$$

$$\left\{\left\{-\frac{2\sqrt{\frac{2}{5}}\,\text{Sin}[t]}{\sqrt{3+\text{Cos}[2\,t]}},\ \frac{3\sqrt{\frac{2}{5}}\,\text{Cos}[t]}{\sqrt{3+\text{Cos}[2\,t]}},\ \frac{\sqrt{\frac{2}{5}}}{\sqrt{3+\text{Cos}[2\,t]}}\right\},\right.$$

$$\left\{-\frac{8\,\text{Cos}[t]}{\sqrt{-(-17+\text{Cos}[2\,t])\,(3+\text{Cos}[2\,t])}},\right.$$

$$\left.-\frac{6\,\text{Sin}[t]}{\sqrt{-(-17+\text{Cos}[2\,t])\,(3+\text{Cos}[2\,t])}},\ \frac{2\,\text{Cos}[t]\,\text{Sin}[t]}{\sqrt{-(-17+\text{Cos}[2\,t])\,(3+\text{Cos}[2\,t])}}\right\},$$

$$\left.\left.\left\{\frac{3\sqrt{\frac{2}{5}}\,\text{Sin}[t]}{\sqrt{17-\text{Cos}[2\,t]}},\ -\frac{2\sqrt{\frac{2}{5}}\,\text{Cos}[t]}{\sqrt{17-\text{Cos}[2\,t]}},\ \frac{6\sqrt{\frac{2}{5}}}{\sqrt{17-\text{Cos}[2\,t]}}\right\}\right\}\right\}$$

```
helix = ParametricPlot3D[r[t], {t, -4 Pi, 4 Pi},
  ColorFunction → Function[{x, y, z}, RGBColor[1, 0, 0]], Axes → False]
```



```
Manipulate[Show[Graphics3D[{Green, PointSize[0.04], Point[r[t]]}], helix],
 {t, 0, 4 Pi}, SaveDefinitions → True]
```



A point moving on a elliptical helix.

Tangent Bundle and Osculating Plane

Tangent bundle

```
tgspace = ParametricPlot3D[r[t] + s r'[t], {t, -4 Pi, 4 Pi}, {s, -1, 1}, Mesh → False,
  ColorFunction → Function[{x, y, z}, Directive[Opacity[0.1], Blue]]]
```



In the picute above a part of the tangent space

```
Manipulate[Show[Graphics3D[{Green, PointSize[0.02], Point[r[t]],
    Line[{r[t] - 10 r'[t] / Norm[r'[t]], r[t] + 10 r'[t] / Norm[r'[t]]}]}],
  helix, PlotRange → {{-5, 5}, {-5, 5}, {-16, 16}}],
 {t, -4 Pi, 4 Pi}, SaveDefinitions → True]
```



```
Manipulate[Show[
  Graphics3D[{Green, PointSize[0.02], Point[r[t]], Arrow[{r[t], r[t] + r'[t]}]}],
  helix, tgspace, PlotRange → {{-5, 5}, {-5, 5}, {-16, 16}}],
 {t, -4 Pi, 4 Pi}, SaveDefinitions → True]
```



The osculating plane:

The osculating plane to a curve *K* at a point *P* is the limit (if it exists) of the planes containing the tangent to *K* at *P* and a point *Q* as *Q* tends to *P*.

equation of osculating plane:

$$\left|\{\{x, y, z\} - r(t), r'(t), r''(t)\}\right| = 0$$

$$-2\, y \operatorname{Cos}[t] - 6\, t \operatorname{Cos}[t]^2 + 6\, z \operatorname{Cos}[t]^2 + 3\, x \operatorname{Sin}[t] - 6\, t \operatorname{Sin}[t]^2 + 6\, z \operatorname{Sin}[t]^2 == 0$$

```
osculating[t_, r_] :=
 First[z /. Solve[Det[{{x, y, z} - r[t], r'[t], r''[t]}] == 0, z]]
```

```
Manipulate[Plot3D[osculating[t, r], {x, -10, 10}, {y, -10, 10}],
 {t, -4 Pi, 4 Pi}, SaveDefinitions → True]
```



```
Manipulate[Show[Graphics3D[{Green, PointSize[0.02], Point[r[t]],
    Line[{r[t] - 10 r'[t] / Norm[r'[t]], r[t] + 10 r'[t] / Norm[r'[t]]}]}],
  helix, Plot3D[osculating[t, r], {x, -10, 10}, {y, -10, 10}],
  PlotRange → {{-5, 5}, {-5, 5}, {-16, 16}}], {t, -4 Pi, 4 Pi}]
```



Tangent, Normal, Binormal vectors

(unit) tangent vector

$$\text{tg[r\_][t\_]} := \frac{r'(t)}{\sqrt{r'(t).r'(t)}}$$

curvature

$$\kappa[r\_][t\_] := \sqrt{\frac{tg[r]'[t].tg[r]'[t]}{r'[t].r'[t]}}$$

$\kappa[r][t]$ // Simplify

$$\frac{2}{5}\sqrt{-\frac{-17+\text{Cos}[2\,t]}{(3+\text{Cos}[2\,t])^3}}$$

(unit) normal vector

```
Clear[nr]
```

$$nr[r\_][t\_] := \frac{tg(r)'(t)}{\sqrt{tg(r)'(t).tg(r)'(t)}}$$

```
nr[r][t] // FullSimplify
```

$$\left\{-\frac{8\,\text{Cos}[t]}{\sqrt{-\frac{-17+\text{Cos}[2\,t]}{(3+\text{Cos}[2\,t])^2}}\,(3+\text{Cos}[2\,t])^{3/2}},\ -\frac{6\,\text{Sin}[t]}{\sqrt{-\frac{-17+\text{Cos}[2\,t]}{(3+\text{Cos}[2\,t])^2}}\,(3+\text{Cos}[2\,t])^{3/2}},\ \frac{\text{Sin}[2\,t]}{\sqrt{-\frac{-17+\text{Cos}[2\,t]}{(3+\text{Cos}[2\,t])^2}}\,(3+\text{Cos}[2\,t])^{3/2}}\right\}$$

binormal vector

```
bi[r_][t_] := Cross[tg[r][t], nr[r][t]]
```

```
Manipulate[Show[
  Graphics3D[{Black, Arrowheads[Small], Arrow[{r[t], r[t] + tg[r][t]}], Brown,
    Arrow[{r[t], r[t] + nr[r][t]}], Blue, Arrow[{r[t], r[t] + bi[r][t]}]}],
  helix], {t, -Pi, Pi}, SaveDefinitions -> True]
```



Torsion

```
bi[r]'[t] / ((Sqrt[r'(t).r'(t)]) nr[r][t]) // Simplify
```

$$\left\{\frac{12}{5\,(-17+\text{Cos}[2\,t])},\ \frac{12}{5\,(-17+\text{Cos}[2\,t])},\ \frac{12}{5\,(-17+\text{Cos}[2\,t])}\right\}$$

```
First[bi[r]'[t]] / First[((Sqrt[r'(t).r'(t)]) nr[r][t])] // Simplify
```

$$\frac{12}{5\,(-17+\text{Cos}[2\,t])}$$

Another definition of torsion

$$\text{torsion}(r\_)(t\_) := -\frac{\text{bi}(r)'(t).\text{nr}(r)(t)}{\sqrt{r'(t).r'(t)}}$$

```
torsion[r][t] // Simplify
```

$$-\frac{12}{5\,(-17+\mathrm{Cos}[2\,t])}$$

```
r1[t_] := {2 Cos[t], 2 Sin[t], t};
```

```
κ[r1][t] // Simplify
```

$$\frac{2}{5}$$

```
torsion[r1][t] // Simplify
```

$$\frac{1}{5}$$

```
FullSimplify[κ[{#, #^2, #^3} &][t]^2, t > 0]
```

$$\frac{4\left(1+9\left(t^2+t^4\right)\right)}{\left(1+4\,t^2+9\,t^4\right)^3}$$

```
FullSimplify[torsion[{#, #^2, #^3} &][t], t > 0]
```

$$\frac{3}{1+9\left(t^2+t^4\right)}$$

Another definition of the osculating plane

```
osculatingPlane[t_, r_] :=
 ParametricPlot3D[Evaluate[r[t] + s tg[r][t] + p bi[r]'[t]], {s, -5, 5}, {p, -5, 5}]
```

```
Manipulate[Show[Graphics3D[{Green, PointSize[0.02], Point[r[t]],
    Line[{r[t] - 10 r'[t] / Norm[r'[t]], r[t] + 10 r'[t] / Norm[r'[t]]}]}],
  helix, osculatingPlane[t, r], PlotRange → {{-5, 5}, {-5, 5}, {-16, 16}}],
 {t, 0, 4 Pi}, SaveDefinitions → True]
```



Serret - Frenet Formulas

Assuming$\left[t > 0, \text{FullSimplify}\left[\frac{\text{tg}(r)'(t)}{\sqrt{r'(t).r'(t)}} == \kappa[r][t]\,\text{nr}[r][t]\right]\right]$

True

Simplify$\left[\frac{\text{nr}[r]'[t]}{\sqrt{r'(t).r'(t)}} == \text{torsion}[r][t]\,\text{bi}[r][t] - \kappa[r][t]\,\text{tg}[r][t],\ t > 0\right]$

True

```
FullSimplify[bi[r]'[t] / Sqrt[r'[t].r'[t]] == -torsion[r][t] nr[r][t], t > 0]
```

```
True
```

The Osculating Circle and The Osculating Sphere

The Osculating Circle

The osculating circle (or circle of curvature) of a curve *C* at a point *P* is a circle in the osculating plane at *P* that has a three-point contact with *C* at *P*. The radius of the osculating circle $\frac{1}{\kappa}$ is the radius of curvature of the curve at *P*.

One way to draw the osculating cicle is to start with an ordinary circle at the origin and then translate and rotate it in the the correct position in the osculating plane.

```
osculatingPlane[t_, r_] := ParametricPlot3D[
  Evaluate[r[t] + s tg[t, r] + p bi[r]'[t]], {s, -10, 10}, {p, -10, 10}]
```

```
gr = Cases[ParametricPlot3D[1 / κ[r][1 / 2] {Cos[θ], Sin[θ], 0}, {θ, 0, 2 Pi}],
    _Line, Infinity][[1]];
```

$$|\{\{x, y, z\} - r(t), r'(t), r''(t)\}|$$

```
-2 y Cos[t] - 6 t Cos[t]^2 + 6 z Cos[t]^2 + 3 x Sin[t] - 6 t Sin[t]^2 + 6 z Sin[t]^2
```

```
Clear[v]
```

```
v[t_] := Last[CoefficientArrays[-2 y Cos[t] - 6 t Cos[t]^2 +
      6 z Cos[t]^2 + 3 x Sin[t] - 6 t Sin[t]^2 + 6 z Sin[t]^2, {x, y, z}] // Normal]
```

```
Manipulate[Show[Graphics3D[{Green, PointSize[0.02], Point[r[t]]}],
  helix, Plot3D[osculating[t, r], {x, -20, 20}, {y, -20, 20}, Mesh → False,
   ColorFunction → Function[{x, y, z}, Directive[Opacity[0.2], Blue]]],
  Graphics3D[GeometricTransformation[
    Cases[ParametricPlot3D[1 / κ[r][t] {Cos[θ], Sin[θ], 0}, {θ, 0, 2 Pi}], _Line,
      Infinity][[1]], {TranslationTransform[r[t] + 1 / κ[r][t] nr[r][t]].
      RotationTransform[{{0, 0, 1}, v[t]}]}]],
  PlotRange → {{-20, 20}, {-20, 20}, {-16, 16}}], {t, -4 Pi,
  4 Pi}, SaveDefinitions → True]
```



The Osculating Sphere

The osculating sphere at a point at a point on a curve is the the sphere which has a four-point contact with the curve at *P*. The center of the osculating sphere is called the center of spherical curvature. Its position vector is given by

$$c(t) = \mathbf{bi}(r)(t)\,\sigma(r)(t)\,\rho(r)'(t) + \mathbf{nr}(r)(t)\,\rho(r)(t) + r(t)$$

where

```
σ[r_][t_] := 1 / torsion[r][t]
ρ[r_][t_] := 1 / κ[r][t]
```

The radius of spherical curvature is $\sqrt{(\sigma(r)(t))^2\,(\rho(r)'(t))^2 + (\rho(r)(t))^2}$.

```
Manipulate[
 Show[Graphics3D[{Green, PointSize[0.02], Point[r[t]], Yellow, Opacity[0.1],
    Sphere[r[t] + ρ[r][t] nr[r][t] + σ[r][t] ρ[r]'[t] bi[r][t],
     Sqrt[ρ[r][t]^2 + σ[r][t]^2 ρ[r]'[t]^2]]}],
  helix, PlotRange → {{-20, 20}, {-20, 20}, {-30, 30}}],
 {t, -4 Pi, 4 Pi}, SaveDefinitions → True]
```



Involutes and Evolutes

involute

An involute of C is a curve which lies on the tangent surfact of C and intersects the generators orthogonally. It’s equation is $R = r + (c - s)\,t$

```
Manipulate[Show[Graphics3D[{Green, PointSize[0.02]}],
  ParametricPlot3D[r[s] + (c - s) tg[r][s], {s, -4 Pi, 4 Pi}],
  helix, PlotRange → {{-20, 20}, {-20, 20}, {-30, 30}}],
 {{c, 1, "parameter"}, -2, 2}, SaveDefinitions → True]
```



Evolute

The evolute of a curve C is a curve whose invoute is C.

```
evolute3D[r_][t_] := r[t] + nr[r][t]/κ[r][t] - κ[r]′[t] bi[r][t]/(Sqrt[r′[t].r′[t]] κ[r][t]^2 torsion[r][t])

h[t_] = FullSimplify[evolute3D[r][t], Assumptions → t > 0]
```

$$\left\{\frac{1}{2}\,\text{Cos}[t]\,(-11+5\,\text{Cos}[2\,t]),\ \frac{1}{3}\,(9+5\,\text{Cos}[2\,t])\,\text{Sin}[t],\ t-15\,\text{Cos}[t]\,\text{Sin}[t]\right\}$$

```
ParametricPlot3D[{h[t], r[t]}, {t, 0, 12 Pi},
 PlotStyle → {Green, Red}, Boxed → False, Axes → False]
```



```
s[t_] := {Cos[t], Sin[t], 2 t}
```

```
g[t_] = FullSimplify[evolute3D[s][t], Assumptions → t > 0]
```

```
{-4 Cos[t], -4 Sin[t], 2 t}
```

```
ParametricPlot3D[{g[t], s[t]},
 {t, 0, 12 Pi}, PlotStyle → {Green, Red}, Axes → False]
```



Built-in function FrenetSerretSystem

```
? FrenetSerretSystem
```

FrenetSerretSystem[{x_1, …, x_n}, t] gives the generalized curvatures and Frenet–Serret basis for the parametric curve x_i[t].
FrenetSerretSystem[{x_1, …, x_n}, t, *chart*] interprets the x_i as coordinates in the specified coordinate chart. >>

```
sf = Assuming[Element[t, Reals], FullSimplify[FrenetSerretSystem[r[t], t]]]
```

$$\left\{\left\{\frac{2\sqrt{17-\text{Cos}[2t]}}{5\,(3+\text{Cos}[2t])^{3/2}},\ -\frac{12}{5\,(-17+\text{Cos}[2t])}\right\},\ \left\{\left\{-\frac{2\sqrt{\frac{2}{5}}\,\text{Sin}[t]}{\sqrt{3+\text{Cos}[2t]}},\ \frac{3\sqrt{\frac{2}{5}}\,\text{Cos}[t]}{\sqrt{3+\text{Cos}[2t]}},\ \frac{\sqrt{\frac{2}{5}}}{\sqrt{3+\text{Cos}[2t]}}\right\},\ \left\{-\frac{8\,\text{Cos}[t]}{\sqrt{-(-17+\text{Cos}[2t])\,(3+\text{Cos}[2t])}},\ -\frac{6\,\text{Sin}[t]}{\sqrt{-(-17+\text{Cos}[2t])\,(3+\text{Cos}[2t])}},\ \frac{\text{Sin}[2t]}{\sqrt{-(-17+\text{Cos}[2t])\,(3+\text{Cos}[2t])}}\right\},\ \left\{\frac{3\sqrt{2}\,\text{Sin}[t]}{\sqrt{85-5\,\text{Cos}[2t]}},\ -\frac{2\sqrt{2}\,\text{Cos}[t]}{\sqrt{85-5\,\text{Cos}[2t]}},\ \frac{6\sqrt{2}}{\sqrt{85-5\,\text{Cos}[2t]}}\right\}\right\}\right\}$$

```
sf[[1, 1]]
```

$$\frac{2\sqrt{17-\text{Cos}[2t]}}{5\,(3+\text{Cos}[2t])^{3/2}}$$

κ[r][t] // FullSimplify

$$\frac{2}{5}\sqrt{-\frac{-17+\text{Cos}[2t]}{(3+\text{Cos}[2t])^{3}}}$$

```
sf[[1, 2]]
```

$$-\frac{12}{5\,(-17+\text{Cos}[2t])}$$

```
torsion[r][t] // Simplify
```

$$-\frac{12}{5\,(-17+\text{Cos}[2t])}$$

```
sf[[2, 1]]
```

$$\left\{-\frac{2\sqrt{\frac{2}{5}}\,\text{Sin}[t]}{\sqrt{3+\text{Cos}[2t]}},\ \frac{3\sqrt{\frac{2}{5}}\,\text{Cos}[t]}{\sqrt{3+\text{Cos}[2t]}},\ \frac{\sqrt{\frac{2}{5}}}{\sqrt{3+\text{Cos}[2t]}}\right\}$$

tg[r][t] // Simplify

$$\left\{-\frac{2\sqrt{\frac{2}{5}}\,\text{Sin}[t]}{\sqrt{3+\text{Cos}[2t]}},\ \frac{3\sqrt{\frac{2}{5}}\,\text{Cos}[t]}{\sqrt{3+\text{Cos}[2t]}},\ \frac{\sqrt{\frac{2}{5}}}{\sqrt{3+\text{Cos}[2t]}}\right\}$$

```
sf[[2, 2]]
```

$$\left\{-\frac{8\,\text{Cos}[t]}{\sqrt{-(-17+\text{Cos}[2t])\,(3+\text{Cos}[2t])}},\ -\frac{6\,\text{Sin}[t]}{\sqrt{-(-17+\text{Cos}[2t])\,(3+\text{Cos}[2t])}},\ \frac{\text{Sin}[2t]}{\sqrt{-(-17+\text{Cos}[2t])\,(3+\text{Cos}[2t])}}\right\}$$

```
Assuming[Element[t, Reals], FullSimplify[nr[r][t] ]]
```

$$\left\{-\frac{8\,\text{Cos}[t]}{\sqrt{-(-17+\text{Cos}[2\,t])\,(3+\text{Cos}[2\,t])}},\ -\frac{6\,\text{Sin}[t]}{\sqrt{-(-17+\text{Cos}[2\,t])\,(3+\text{Cos}[2\,t])}},\ \frac{\text{Sin}[2\,t]}{\sqrt{-(-17+\text{Cos}[2\,t])\,(3+\text{Cos}[2\,t])}}\right\}$$

```
sf[[2, 3]]
```

$$\left\{\frac{3\sqrt{2}\,\text{Sin}[t]}{\sqrt{85-5\,\text{Cos}[2\,t]}},\ -\frac{2\sqrt{2}\,\text{Cos}[t]}{\sqrt{85-5\,\text{Cos}[2\,t]}},\ \frac{6\sqrt{2}}{\sqrt{85-5\,\text{Cos}[2\,t]}}\right\}$$

```
Assuming[Element[t, Reals], FullSimplify[bi[r][t]]]
```

$$\left\{\frac{3\sqrt{2}\,\text{Sin}[t]}{\sqrt{85-5\,\text{Cos}[2\,t]}},\ -\frac{2\sqrt{2}\,\text{Cos}[t]}{\sqrt{85-5\,\text{Cos}[2\,t]}},\ \frac{6\sqrt{2}}{\sqrt{85-5\,\text{Cos}[2\,t]}}\right\}$$